ARDUINO
STRUTTURA PROGRAMMA
Il sorgente (detto Sketch) è formato da tre parti fisse :

1. dichiarazione delle variabili (come nel C++)

2. settaggio delle impostazioni hardware utilizzando la funzione : void setup(). Questa funzione verrà eseguita una volta all'avvio

3. il programma vero e proprio utilizzando la funzione : void loop(). Questa funzione verrà eseguita fino allo spegnimento della scheda.

COSTANTI DI SISTEMA
HIGH e LOW definiscono il livello logico rispettivamente alto e basso. Per esempio per definire lo stato logico di un pin digitale di uscita

INPUT e OUTPUT definiscono se un pin I/O è ingresso o uscita

TRUE o FALSE è la conseguenza dell'analisi di qualsiasi espressione che può essere vera o falsa.

TIPI DI VARIABILE
boolean : possono essere vere o false. Es.  boolean pippo = true ; //la variabile pippo vale “1”.
char : contiene un carattere salvato come numero (ASCII). Se usato come numero può memorizzare valori da 

-128 a 127.
byte : 1 B di memoria RAM. Può contenere valori da 

0 a 255.
int : 2 B di memoria RAM. Può contenere valori da 

-32 768 a 32 767.

unsigned : 2 B di memoria RAM. Può contenere valori da 0 a 65 535.
long : 4 B di memoria RAM. Può contenere valori da 

-2 147 483 648 a 2 147 483 647.
unsigned long : 4 B di memoria RAM. Può contenere valori da 

0 a 4 294 967 295 
float : 4 B di memoria RAM. Può contenere numeri con virgola.
Double : a doppia precisione in virgola mobile. Il valore massimo è 

17 976 931 348 623 157*10^308
string : array formato da caratteri ASCII per memorizzare testi. Ogni carattere è 1 B di RAM e termina sempre con il carattere null per indicare il termine della stringa. 

Es. char pippo[] = “Ciao”; è una stringa formata da 5 caratteri, 4 della parola memorizzata e il null di chiusura.
Array elenco di valori dello stesso tipo identificabili con un indice. 

Es. int pippo[5]={3,124,67,12}. Come per la string, che è un tipo di array per memorizzare caratteri, l'ultimo dato è null per indicare al sistema che i dati sono terminati.

I/O DIGITALE

livelli lettura

· “0” va da 0 a 2,5 V

· “1” va da 3 a 5 V (Vcc)

Per ottenere l'uscita tre-state : si setta il pin come ingresso e non lo si legge.

E' anche possibile attivare una R di pull-up internaquando il pin è settato come ingresso digitale.

Il pin 13 ha il led “L” sulla scheda che permette di monitorare lo stato logico o comunque l'uscita.

I/O ANALOGICO

Gli ingressi analogici sono siglati A0, A1, A2, A3, A4 e A5 (per quanto riguarda la scheda Arduino Uno Rev.3)

Il micro ha all'interno un solo convertitore A/D di tipo S&H. Quindi può essere letto un singolo pin alla volta.
AREF : a questo pin va applicato il valore Vmax di riferimento per i pin analogici.

Il convertitore ha una sensibilità di 48 mV e divide l'asse della tensione in 1024 gradini.

La frequenza massima di campionamento è 4,5 Khz.

Le uscite non sono analogiche ma sono delle uscite PWM (pulse width modulation). Le frequenze dei pin sono fisse :

· pin 5 e 6 : 976 Hz

· tutti gli altri pin PWM : 488 Hz

Queste frequenze possono essere modificate agendo sul prescaler interno del microcontrollore.
La formula per calcolare il valore dei componenti della rete RC è 
[image: image3.emf][Hz]
se il circuito a valle ha un assorbimento troppo alto da provocare l'abbassamento della tensione, bisogna inserire un operazionale con schema da inseguitore.

[image: image7.png]Vou




INTERRUPT

pin 2 e 3 (sulla scheda Arduino Uno Rev.3)

la funzione richiede 4 tipi di funzionamento :

1. LOW : si attiva quando il pin specificato va a livello basso

2. CHANGE : si attiva quando il pin cambia stato

3. RISING : legge la fase di salita, quindi il passaggio da “0” a “1”

4. FALLING : legge la fase di discesa, quindi il passaggio da “1” a “0”

STRUTTURE DI CONTROLLO
if

if … else

for

switch
la struttura è
switch (variabile)

{

case: 
[image: image4.emf]; se la variabile vale n allora viene eseguito il blocco sotto, altrimenti va avanti

.istruzione 1;

istruzione 2;

.

istruzione m;

break;

case : 
[image: image5.emf];

.

.

.

default :

istruzione 1;

istruzione 2;

.

istruzione m;

}

e così via per tutti i valori da controllare.
while

do … while

break : interrompe un ciclo e prosegue il programma all'esterno del ciclo.
continue : fa saltare le istruzioni successive del ciclo, pero il ciclo poi prosegue con i valori seguenti.
Return : serve per restituire un valore di una funzione. Si possono usare più return in una funzione, ma il primo che viene eseguito chiude la funzione chiamata con il risultato corrente.
goto
OPERAZIONI ARITMETICHE O DI COMPARAZIONE O OPERATORI BOOLEANI
Come il c++.

FUNZIONI INPUT E OUTPUT
pinMode(pin,mode)
serve per impostare un certo pin digitale come ingresso o uscita :

pin : numero del pin (scritto sulla scheda Arduino)

mode : INPUT o OUTPUT (rispettivamente ingresso o uscita)

es - pinmode(4,OUTPUT); // battezza il pin 4 come uscita

digitalWrite(pin,value)
definisce lo stato logico di un pin impostato come uscita digitale :

pin : numero del pin (scritto sulla scheda Arduino)

value : HIGH o LOW (rispettivamente “1” o “0”)

digitalRead(pin)
legge lo stato logico del pin impostato come ingresso digitale :

pin : numero del pin (scritto sulla scheda Arduino)

es – pippo=digitalRead(4); assegna alla variabile pippo il valore HIGH  (“1”) o LOW (“0”)

analogWrite(pin,value)
varia il duty-cycle sul pin indicato e mediante un filtro passa basso si può trasformare in una tensione di uscita continua variabile.

pin : numero del pin (scritto sulla scheda Arduino). [A0, A1, A2, A3, A4 e A5]

value : 0 … 255 (rappresenta il 0 … 100 % del duty-cycle)

analogRead(pin)
legge la tensione applicata al pin specificato e lo memorizza con un valore da 0 a 1023 per una tensione di ingresso che va da 0 a 5 V. La sensibilità è di 48 mV.

pin : numero del pin (scritto sulla scheda Arduino). [A0, A1, A2, A3, A4 e A5]

attachInterrupt(interrupt, function, mode)
gestisce l'interrupt.

Interrupt : 0 lo lega al pin 2, 1 lo lega al pin 3 (scheda Arduino Uno Rev.3)

function : la funzione da chiamare quando viene chiamato questo interrupt

mode : modalità di funzionamento
LOW : si attiva quando il pin specificato va a livello basso

CHANGE : si attiva quando il pin cambia stato

RISING : legge la fase di salita, quindi il passaggio da “0” a “1”

FALLING : legge la fase di discesa, quindi il passaggio da “1” a “0”
Sorgente base : sorgente da utilizzare per cominciare

//SORGENTE BASE per scheda Arduino Uno Rev.3

//***********************************************************************************

//COGNOME :

//NOME :

//CLASSE :

//SEZIONE :

//INDIRIZZO :

//A.S. :

//***********************************************************************************

//DICHIARARE LE VARIABILI  :  tipo nome_variabile;

//se si vuole assegnare un valore alla variabile : tipo nome_variabile=valore;

int led = 13;

int pul=7;

// IMPOSTAZIONE HARDWARE (viene eseguito una volta all'accensione o dopo RESET

void setup() 

{                

//INPUT DIGITALI : pinMode(pin,INPUT);

  pinMode(pul,INPUT);

//OUTPUT DIGITALI : pinMode(pin,OUTPUT);

  pinMode(led, OUTPUT); 

//INPUT ANALOGICI : nome_variabile=AnalogRead(pin);  

//pin : A0, A1, A2, A3, A4, A5

//OUTPUT ANALOGICI (PWM) : analogWrite(pin,value)   

//si possono usare i pin contrassegnati da ~ (tilde) : 3 5 6 9 10 11

// PROTOTIPI FUNZIONI

//INTERRUPT : attachInterrupt(0/1, nome_funzione, mode)   

// mode : LOW, CHANGE, RISING, FALLING

//ALTRO .....

}

// SORGENTE RIPETUTO ALL'INFINITO (cioè fino a quando si spegne la scheda o si resetta)

void loop() 

{

//ISTRUZIONI 

}

//FUNZIONI 

FUNZIONI UTILI
millis(x);

micros(x);

delay(x);

delayMicroseconds(x);

min(x,y);

max(x,y);

abs(x);

contrain(x,a,b); // ritorna il valore di x, ma sotto a da a, sopra b da b
map(value, fromlow, fromhigh, tolow, tohigh);//value può essere variabile o funzione. Il valore della variabile di range fromlow – fromhigh verrà traslato ad un range tolow – tohigh.
double pow(b,e);

double sqrt(x);

double sin(rad);

double cos(rad);

double tan(rad);

random(min,max);

randomSeed();

SERIALE - USB
Serial.begin(speed)

Serial.print(data,codifica) [codifica è opzionale : specifica il base numerica (per i tipi di dati integrali) o il numero di cifre decimali (per i tipi di virgola mobile)]
es. 

Serial.print(45); // invia 45

Serial.print(45, DEC); // invia 45 in decimale come sopra
Serial.print (78, BIN) dà "1.001.110"
Serial.print (78, OCT) dà "116"

Serial.print (78, DEC) dà "78"

Serial.print (78, HEX) dà "4E"

Serial.println (1,23456, 0) dà "1"

Serial.println (1,23456, 2) dà "1.23"

Serial.println (1,23456, 4) dà "1,2346"

Serial.print(32, BYTE); //”space” nel codice ASCII-encoded
Serial.available(); // ritorna quanti bytes sono sulla porta seriale non ancora letti

Serial.read(); //legge byte in entrata sulla seriale

Poichè il programma potrebbe essere lento a leggere i dati che arrivano, questi vengono memorizzati in un buffer. Per pulire questo buffer, al termine della lettura :
Serial.flush();
/*
Uses a FOR loop for data and prints a number in various formats.
*/
int x = 0;    // variable

void setup() {
  Serial.begin(9600);      // open the serial port at 9600 bps:    
}

void loop() {  
  // print labels 
  Serial.print("NO FORMAT");       // prints a label
  Serial.print("\t");              // prints a tab

  Serial.print("DEC");  
  Serial.print("\t");      

  Serial.print("HEX"); 
  Serial.print("\t");   

  Serial.print("OCT");
  Serial.print("\t");

  Serial.print("BIN");
  Serial.print("\t"); 

  for(x=0; x< 64; x++){    // only part of the ASCII chart, change to suit

    // print it out in many formats:
    Serial.print(x);       // print as an ASCII-encoded decimal - same as "DEC"
    Serial.print("\t");    // prints a tab

    Serial.print(x, DEC);  // print as an ASCII-encoded decimal
    Serial.print("\t");    // prints a tab

    Serial.print(x, HEX);  // print as an ASCII-encoded hexadecimal
    Serial.print("\t");    // prints a tab

    Serial.print(x, OCT);  // print as an ASCII-encoded octal
    Serial.print("\t");    // prints a tab

    Serial.println(x, BIN);  // print as an ASCII-encoded binary
    //                             then adds the carriage return with "println"
    delay(200);            // delay 200 milliseconds
  }
  Serial.println("");      // prints another carriage return
}




Adesso collegate il vostro Arduino al computer e caricate il seguente sketch:
/* PROGRAMMA DI PROVA DI INTERFACCIA SERIALE BIDIREZIONALE
 Il programma riceve comandi dalla seriale e spedisce
 letture e stati sullo stesso canale.
 Per provare lo sketch, collegare un LED al pin 9
 con un resistore adeguato ed una tensione in ingresso
 al pin A0.
 Caricato lo sketch sull'Arduino, aprire il monitor seriale
 dell'IDE di Arduino con velocità di 19200 bps.
 I comandi accettati sono i seguenti:
 l - lettura: legge il valore della tensione in ingresso al pin analogico
 a - accedi LED: accende il LED
 s - spengi LED: spenge il LED
 q - richiede lo stato del LED. Verrà spedito lo stato del pin
 a cui è collegato il LED. Possibili risposte:
 LED acceso/spento/pilotato con PWM impostato a xxx
 w xxx - scrive un valore di PWM sul pin 9. Possibili
 valori per "xxx" vanno da 0 a 255. Gli spazi sono ignorati.
 I caratteri accettati sono solo quelli numerici (da 0 a 9).
 */
 //pin
 const byte inputPin = A0; //pin analogico di ingresso
 const byte outputPin = 9; //pin di output
 //variabili globali
 byte statoLed = 0; //contiene il valore sul pin 9
 void setup() {
   Serial.begin(19200); // apro la seriale
   delay(1000); //attendo che l'utente faccia altrettanto
   Serial.println(F("Pronto")); //sono pronto
 }
 //programma principale
 void loop() {
   int lettura; //contiene il valore di PWM letto dalla seriale
   unsigned long tempMillis; //serve per il timeout sulla ricezione del valore di PWM
   byte caratteri; //contiene il numero di caratteri ricevuti
   byte tempChar; //contiene il carattere letto
   //controllo se c'è qualcosa in arrivo sulla seriale
   if (Serial.available()) {
     byte command = Serial.read(); //leggo il primo byte
     switch (command) { //controllo che sia un comando valido
       case 'l': //lettura
         //invio la lettura dell'ingresso
         Serial.print(F("Lettura ingresso: "));
         Serial.println(analogRead(inputPin));
         break;
       case 'a': //accendo il LED
         statoLed = 255;
         analogWrite(outputPin, statoLed);
         Serial.println(F("LED acceso"));
         break;
       case 's': //spengo il led
         statoLed = 0;
         analogWrite(outputPin, statoLed);
         Serial.println(F("LED spento"));
         break;
       case 'q': //l'utente vuole lo stato del LED
         Serial.print(F("LED "));
         if (statoLed == 0) {
           Serial.println(F("spento"));
         } else if (statoLed == 255) {
           Serial.println(F("acceso"));
         } else {
           Serial.print(F("pilotato con PWM impostato a "));
           Serial.println(statoLed, DEC);
         }
         break;
       case 'w': //imposto il PWM sul LED
         lettura = 0;
         tempMillis = millis();
         caratteri = 0;
         //mi servono altri caratteri
         do {
           if (Serial.available()) {
             tempChar = Serial.read();
             caratteri++;
             if ((tempChar >= 48) && (tempChar <= 57)) { //è un numero? ok
               lettura = (lettura * 10) + (tempChar - 48);
             } else if ((tempChar == 10) || (tempChar == 13)) {
               //con invio e/o nuova linea esco
               break;
             }
           }
         //esco per timeout, per un valore di PWM maggiore di 255
         //oppure per troppi caratteri senza senso ricevuti
         } while((millis() - tempMillis < 500) && (lettura <= 255) && (caratteri < 10));
         //se il valore di PWM è valido, lo imposto sul pin
         if (lettura <= 255) {
           statoLed = lettura;
           analogWrite(outputPin, statoLed);
           Serial.print(F("LED acceso con PWM impostato a "));
           Serial.println(statoLed, DEC);
         }
         break;
     }
     //svuoto il buffer da eventuali altri caratteri che non mi servono più
     while (Serial.available()) {
       byte a = Serial.read();
     }
   }
 }
Adesso aprite il monitor seriale dell’IDE (cliccate sull’icona a forma di lente d’ingrandimento posta in alto a destra, sopra all’area di editing del codice) con una velocità di comunicazione di 19200 bps (la stessa impostata nello sketch). Dovreste ricevere il messaggio “Pronto”: questo significa che lo sketch è partito e che sta attendendo i vostri comandi.

Adesso interagiamo con l’Arduino usando i seguenti comand (i comandi vanno inseriti nel monitor e spediti all’Arduino premendo il tasto “invio”):
· q – questo comando chiede all’Arduino lo stato del LED. L’Arduino risponderà con LED acceso oppure spento nel caso che il LED sia acceso con la massima intensità oppure spento, oppure risponderà con il valore corrente del segnale PWM (da 0 a 255).
· a – questo comando accende il LED alla massima intensità.
· s – questo comando spenge il LED.
· l – questo comando fa leggere all’Arduino la tensione sul pin A0 e poi spedisce la lettura al PC: vedrete un messaggio del tipo “Lettura ingresso: xxx”, con xxx che rappresenta il valore sul pin convertito in formato digitale a 10 bit (da 0 a 1023).
· w numero – con il comando w si spedisce all’Arduino un numero indicante il valore di PWM da assegnare al pin. Il numero può assumere un valore da 0 a 255: 0 equivale al comando “spento” mentre 255 equivale al comando “acceso”. Gli spazi sono ignorati (“w10″ equivale a “w 10″) così come eventuali caratteri diversi dalle cifre numeriche.

Analizziamo il codice, per capire come funziona il tutto.

Sezioni “pin” e “variabili globali”: poco da dire, qui si dichiarano solo le costanti usate per indicare i pin ed alcune variabili usate più avanti.

Funzione “setup()”: apriamo la seriale, attendiamo un secondo per dare modo all’utente di aprire il terminale, poi spediamo il messaggio “Pronto”.

Funzione “loop()”: qui risiede il cuore di tutto. La prima cosa che facciamo è controllare se nel buffer di ricezione della seriale è presente qualcosa, ossia il codice vuol capire se dal computer all’Arduino è stato spedito qualche carattere. La comunicazione seriale è bidirezionale, quindi possiamo spedire dati dal computer all’Arduino e viceversa. L’Arduino ha 2 buffer seriali (ossia due array usati per salvare i dati in arrivo ed in uscita): la funzione Serial.available() resistuisce zero se nessun carattere è arrivato oppure il numero di caratteri ricevuti. Il test if (Serial.available()) serve quindi per capire se c’è almeno un carattere nel buffer perché qualsiasi valore maggiore di zero farà risultare vero il test ed il codice entrerà nel blocco di codice dell’if.

Una volta accertata la presenza di qualcosa nel buffer, preleviamo il primo byte ricevuto.

[notice]Attenzione, parliamo di “byte”, non di caratteri, valori, numeri, stringhe ecc… : la trasmissione seriale avviene spedendo singoli byte, uno dietro l’altro. Ogni byte viene interpretato per come esso viene spedito. Ad esempio, se noi scriviamo Serial.print(“A”) verrà inviato il carattere A, che corrisponde al valore ASCII 65. Questo significa che sulla seriale viaggerà il byte 65. In ricezione, se noi tratteremo quel byte come un carattere, esso corrisponderà alla lettera “A” mentre se noi tratteremo quel byte come valore, esso corrisponderà a 65. [/notice]

Analizziamo il byte per capire che cosa è arrivato usando uno switch..case.

Nel caso di comandi senza parametri (a, s, q, l) l’interpretazione è facile perché non necessitiamo di informazioni aggiuntive: quel comando va eseguito subito. Lo sketch compie quindi l’operazione richiesta ed informa l’utente di ciò che è avvenuto. Questo messaggio di risposta lo vedremo stampato direttamente sul monitor seriale.

Nel caso del comando “w”, invece, necessitiamo del parametro numerico con cui regolare il PWM sul pin 9 dell’Arduino. Come abbiamo detto precedentemente, l’informazione che viaggia sulla seriale è composta da singoli byte: se volessimo regolare il PWM a 100, potremmo semplicemente spedire un byte di valore 100, e l’Arduino non dovrebbe far altro che prendere quel byte e passarlo alla funzione analogWrite. Ma ciò creerebbe un pò di problemi all’utente: il valore ASCII di 100 è la lettera “d”, ci vuole una tabella ASCII per ritrovare il carattere corrispondente al valore 100. E cosa succederebbe se dovessimo spedire il valore 200? Siccome i caratteri ASCII sono standard solo fino al valore 127, da 128 a 255 ognuno riempie la tabella come vuole. Dovremmo trovare una combinazione di tasti sulla tastiera che nella mappatura usata dal nostro sistema crea il carattere con valore 200. Troppo complicato! L’alternativa? Spedire il numero come stringa, ossia inviare il carattere ASCII delle cifre che compongono il numero. 200 viene quindi scomposto come ‘2’, ‘0’, ‘0’ (notare gli apici, che in C indicano che il carattere alfanumerico racchiuso fra di essi deve essere trattato come carattere). E’ in questo modo che lavora il monitor seriale, fatto apposta per favorire l’interazione dell’uomo con la macchina. Quindi quando noi scriviamo “w 200″ il monitor seriale invia all’Arduino 5 caratteri: ‘w’, ‘spazio’, ‘2’, ‘0’, ‘0’. Il nostro codice non deve fare altro che leggere i byte ricevuti e trattarli come caratteri. Per far ciò basta ricordarsi che un carattere ha un preciso codice ASCII, quindi lo spazio corrisponde al codice ASCII 32 per cui questo sarà il byte ricevuto, il carattere ‘2’ ha il codice ASCII 50 per cui questo sarà il byte ricevuto, ecc…

Il ciclo do…while esegue proprio il controllo in questa forma: sui valori ASCII dei caratteri ricevuti. Quindi per scartare lo spazio, basta controllare che il byte abbia valore 32. Per verificare che il carattere ricevuto sia numerico, basta controllare che sia compreso fra i valori di 48 e 57, che corrispondono ai codici ASCII rispettivamente di 0 e 9. Nel programma avremmo potuto sostituire i controlli basati sui numeri sui controlli basati sui caratteri. Quindi
if ((tempChar >= 48) && (tempChar <= 57))
può essere sostituito dal seguente codice:
if ((tempChar >= '0') && (tempChar <= '9'))
senza che il risultato cambi perché il compilatore crea lo stesso identico codice eseguibile. Nel do..while ho introdotto anche un timeout, per evitare che il codice possa bloccarsi per qualche ragione, come ad esempio un’interruzione della linea durante la trasmissione. Il codice accetta solo numeri ed esce non appena si verifica una delle seguenti condizioni: timeout; un valore numerico superiore a 255; la ricezione di più di 10 caratteri insignificanti. La conversione da carattere a valore numerico è fatta usando ancora i codici ASCII dei caratteri: sappiamo che 48 corrisponde a 0, 49 ad 1, 50 a 2 ecc… per cui per avere il valore numerico del carattere ricevuto basta sottrarre appunto 48, per cui: (48-48)=0, (49-48)=1, (50-48)=2 ecc…

Se il valore numerico è corretto (rientra nell’intervallo 0..255), lo scriviamo sul pin PWM con la funzione analogWrite e poi lo salviamo nella variabile che contiene lo stato del pin 9. Terminata quest’operazione, svuotiamo il buffer di ricezione da eventuali altri caratteri che potrebbero essere arrivati (prendiamo il caso di un utente “burlone” che spedisce “w fdsahjawknglkangw”). Sulle vecchie versioni dell’IDE c’era la funzione Serial.flush(), che svuotava il buffer. Con l’introduzione dell’IDE 1.0 questa funzione è stata modificata (adesso attende che i dati in uscita siano stati trasmessi tutti) per cui non resta che fare un ciclo che preleva dal buffer di ricezioni i caratteri uno ad uno, salvandoli in una variabile locale che verrà distrutta non appena usciti dal ciclo do..while.

Ecco un esempio di output:


In un prossimo articolo vedremo come creare un protocollo di trasmissione per lo scambio di dati fra 2 Arduino (o 2 microcontrollori) collegati via seriale tra di loro.

